Calculus Single Variable
Brian E. Blank and Steven G. Krantz

Section 8.5
Work

1. Diminishing Loads: Exercises from the Text

Exercise 1 Page 601

A steam shovel lifts a 500 pound load of gravel from the ground to a point 80 feet above the ground. However the gravel is fine, and it leaks from the shovel at the rate of 1 pound per second. If it takes the steam shovel one minute to lift its load at a constant rate, then how much work is performed?.

Solution

At time t seconds, the height y of the load is $\frac{80\,t}{60}$ feet. Solving for t in terms of y, we have $t = \frac{3\,y}{4}$.

At that time, the load is $500 - t$ pounds. Therefore, at height y, the load is $500 - \frac{3\,y}{4}$ pounds.

Divide the interval $[0,80]$ into N equal subintervals of length Δy. Choose a point y_j in the j^{th} subinterval Lifting the load through the j^{th} subinterval requires about $\left(500 - \frac{3\,y_j}{4}\right)\Delta y$ foot-pounds of work. The total work is about $\sum_{j=1}^{N}\left(500 - \frac{3\,y_j}{4}\right)\Delta y$ foot-pounds. As N increases the approximation becomes more accurate. Letting N tend to infinity we obtain the work as the limit of the sums, which we recognize as Riemann sums:

$$\text{Work} = \lim_{N\to\infty} \sum_{j=1}^{N}\left(500 - \frac{3\,y_j}{4}\right)\Delta y = \int_0^{80} 500 - \frac{3\,y}{4}\,dy = 37600 \text{ foot-pounds.}$$

Ω

Exercise 2 Page 601

When a mass M measured in slugs is y feet above the surface of the earth, its weight in pounds is

$$1.4077 \times 10^{16} \times \frac{M}{(2.0917 * 10^7 + y)^2}.$$

(One pound of force accelerates one slug one foot per second squared.) How much work is done lifting a satellite from the surface of the earth, where it weighs 800 pounds, to a height of 200 miles?

Solution

According to Newton's law, the force of gravity is $F = \frac{G\,M\,m}{(R+y)^2}$ where G is a universal constant, M is the mass of the earth, m is the mass of the satellite, R is the radius of the earth, and y is the height of the satellite above the earth's surface.

```
> R := 6375.58*km;  #Earth's radius in kilometers (given on page 598)
```

$$R := 6375.58\ km$$

```
> R := convert(R, units, km, ft)*ft/km;
```

$$R := 0.2091725722\ 10^{8}\ ft$$

The value of $G\,M$, as measured in the metric unit $\frac{newton\ meter^2}{kilogram}$, is given to be 3.98621×10^{14} (see page 598). Writing the newton in terms of the units for length, mass, and time, namely

$$1\ newton = 1\ kg \times \frac{1\ m}{second^2},$$

we have

$$G\,M = 3.98621 \times 10^{14}\ \frac{meter^3}{second^2}.$$

Thus, in terms of the unit $\frac{ft^3}{second^2}$, $G\,M$ is

```
> eqn1 := G*M[e] = evalf(3.98621*10^14*convert(1, units, m, ft)^3
)*ft^3/s^2;
```

$$eqn1 := G\,M_e = \frac{0.1407716776\ 10^{17}\ ft^3}{s^2}$$

Solve for Newton's constant G:

```
> eqn2 := G = solve(eqn1, G);
```

$$eqn2 := G = \frac{0.1407716776\ 10^{17}\ ft^3}{M_e\,s^2}$$

Substitute into Newton's Law of Gravitation:

```
> NewtonsLaw := F = G*M[e]*M[s]/(R+y)^2;
```

$$NewtonsLaw := F = \frac{G\,M_e\,M_s}{(0.2091725722\ 10^8\ ft + y)^2}$$

```
> eqn3 := subs(eqn2,NewtonsLaw);
```

$$eqn3 := F = \frac{0.1407716776\ 10^{17}\ ft^3\,M_s}{s^2\,(0.2091725722\ 10^8\ ft + y)^2}$$

Let us now find the mass of the satellite. The unit will be the slug. $\left(\ 1\ pound = \frac{1\ slug\ 1\ ft}{s^2}\ \right)$

```
> eqn4 := subs({F=800*slug*ft/s^2, y=0}, eqn3);
```

$$eqn4 := \frac{800\ slug\ ft}{s^2} = \frac{32.17405592\ ft\,M_s}{s^2}$$

```
> eqn5 := M[s] = solve(eqn4, M[s]);
```

$$eqn5 := M_s = 24.86475445\ slug$$

The weight of the satellite y feet above the surface of the earth becomes

```
> eqn6 := subs(eqn5, eqn3);
```

$$eqn6 := F = \frac{0.3500253197\ 10^{18}\ ft^3\ slug}{s^2\ (0.2091725722\ 10^8\ ft + y)^2}$$

To write this in terms of the standard unit of force, the pound, we convert:

```
> eqn7 := subs(slug = pound*s^2/ft, eqn6);
```

$$eqn7 := F = \frac{0.3500253197\ 10^{18}\ ft^2\ pound}{(0.2091725722\ 10^8\ ft + y)^2}$$

```
> Work := Int(rhs(eqn7), y = 0 .. 200*5280*ft);
```

$$Work := \int_0^{1056000\ ft} \frac{0.3500253197\ 10^{18}\ ft^2\ pound}{(0.2091725722\ 10^8\ ft + y)^2}\, dy$$

```
> Work := student[changevar](y=u*ft, Work, u);
```

$$Work := \int_0^{1056000} \frac{0.3500253197\ 10^{18}\ ft\ pound}{(0.2091725722\ 10^8 + u)^2}\, du$$

```
> Work = value( Work );
```

$$\int_0^{1056000} \frac{0.3500253197\ 10^{18}\ ft\ pound}{(0.2091725722\ 10^8 + u)^2}\, du = 0.8042002477\ 10^9\ ft\ pound$$

Ω

Exercise 3 Page 601

A rocket climbs straight up. Its initial total weight, including fuel, is 7000 pounds. Fuel is consumed at the constant rate of 30 pounds per mile. Taking into account the decrease in weight due to fuel consumption, but disregarding the decrease in weight due to increasing elevation, approximate the work done in lifting the rocket the first 20 miles into space.

Solution

At height y measured in feet, the load is $7000 - \frac{30\,y}{5280}$ pounds. Divide the interval $[0, 20(5280)]$ into N equal subintervals of length Δy. Choose a point y_j in the j^{th} subinterval Lifting the load through the j^{th} subinterval requires about $\left(7000 - \frac{30\,y_j}{5280}\right)\Delta y$ foot-pounds of work. The total work is about $\sum_{j=1}^{N}\left(7000 - \frac{30\,y}{5280}\right)\Delta y$ foot-pounds. As N increases the approximation becomes more accurate. Letting N tend to infinity we obtain the work as the limit of the sums, which we recognize as Riemann sums:

$$\text{Work} = \lim_{N\to\infty}\sum_{j=1}^{N}\left(7000 - \frac{30\,y}{5280}\right)\Delta y = \int_0^{20\,(5280)} 7000 - \frac{30\,y}{5280}\,dy = 707{,}520{,}000$$

foot-pounds.

Ω

Exercise

A rocket climbs straight up. Its initial total weight, including fuel, is 7000 pounds. The rate at which the *mass* of fuel is burned is constant. At the surface of the earth the burn rate, stated in pounds, is 30 pounds per mile Taking into account the decrease in weight due to fuel consumption *and* the decrease in weight due to increasing elevation, calculate the work done in lifting the rocket the first 20 miles into space.

Solution

Let M_R denote the mass of the rocket (including fuel) in terms of slugs. Then the weight in pounds is given by

```
> restart:
> NewtonsLaw1 := Weight =
  .1407716776e17*ft^3/s^2*M[R]/(20917257.22*ft+y)^2;
```

$$NewtonsLaw1 := Weight = \frac{0.1407716776\ 10^{17}\ ft^3\ M_R}{s^2\ (0.2091725722\ 10^8\ ft + y)^2}$$

We use this to solve for the initial mass of the fuel/rocket system:

```
> eqn1 := M[R] = solve(subs({y=0*ft, Weight = 7000*slug*ft/s^2},
  NewtonsLaw1), M[R]);
```

$$eqn1 := M_R = 217.5666014\ slug$$

Thus, initially, $M_R = 217.5666014\ slug$.

Now let M_{fuel} be the mass of fuel burned each mile.

```
> NewtonsLaw2 := Weight =
  .1407716776e17*ft^3/s^2*M[fuel]/(20917257.22*ft+y)^2;
```

$$NewtonsLaw2 := Weight = \frac{0.1407716776\ 10^{17}\ ft^3\ M_{fuel}}{s^2\ (0.2091725722\ 10^8\ ft + y)^2}$$

```
> eqn2 := M[fuel] = solve(subs({y=0*ft, Weight = 30*slug*ft/s^2},
  NewtonsLaw2), M[fuel]);
```

$$eqn2 := M_{fuel} = 0.9324282917\ slug$$

The mass of the fuel/rocket system y feet above the surface of the earth is

```
> eqn3 := M[R] = rhs(eqn1)-rhs(eqn2)*y/5280/ft;
```

$$eqn3 := M_R = 217.5666014\ slug - \frac{0.0001765962674\ slug\ y}{ft}$$

The weight of the fuel/rocket system y feet above the surface of the earth becomes

```
> eqn4 := subs(eqn3, NewtonsLaw1);
```

$$eqn4 := Weight = \frac{0.1407716776\ 10^{17}\ ft^3 \left(217.5666014\ slug - \frac{0.0001765962674\ slug\ y}{ft}\right)}{s^2\ (0.2091725722\ 10^8\ ft + y)^2}$$

To write this in terms of the standard unit of force, the pound, we convert:

```
> eqn5 := subs(slug = pound*s^2/ft, eqn4);
```

$$eqn5 := Weight = \frac{0.1407716776\ 10^{17}\ ft^3 \left(\frac{217.5666014\ pound\ s^2}{ft} - \frac{0.0001765962674\ pound\ s^2\ y}{ft^2}\right)}{s^2\ (0.2091725722\ 10^8\ ft + y)^2}$$

```
> Work := Int(rhs(eqn5), y = 0*ft .. 20*5280*ft);
```

$$Work := \int_0^{105600\ ft} \frac{0.1407716776\ 10^{17}\ ft^3 \left(\frac{217.5666014\ pound\ s^2}{ft} - \frac{0.0001765962674\ pound\ s^2\ y}{ft^2}\right)}{s^2\ (0.2091725722\ 10^8\ ft + y)^2}\, dy$$

```
> Work := student[changevar](y=u*ft, Work, u);
```

$$Work := \int_0^{105600} -\frac{0.1407716776\ 10^{17}\ ft\ pound\ (-217.5666014 + 0.0001765962674\ u)}{(0.2091725722\ 10^8 + u)^2}\, du$$

```
> Work := value( Work );
```

$$Work := 0.704018971\ 10^9\ ft\ pound$$

Ω

2. Heavy Cables: Exercises from the Text

Exercise 4 Page 601

A man stands at the top of a tall building and pulls a chain up the side of the building. The chain is 50 feet long and
weighs 3 pounds per linear foot. How much work does the man do in pulling the chain to the top?

Solution

Orient the vertical axis so that the origin is at the top of the building and the positive direction is downward. The chain is realized as the interval [0,50]. Since the weight density is $\frac{3\ pounds}{ft}$, an infinitesmal section of the cable of length $dy\ ft$ weighs 3 $dy\ pounds$. If that section is y feet from the top, the work done lifting it is $y\ 3\ dy\ ft\ pounds$. The total work W is

```
> W := Int(y*3*pounds/ft, y=0*ft..50*ft);
```

$$W := \int_0^{50\ ft} \frac{3\ y\ pounds}{ft}\, dy$$

```
> W = value(W);
```

$$\int_0^{50\ ft} \frac{3\ y\ pounds}{ft}\, dy = 3750\ ft\ pounds$$

Ω

Exercise 6 Page 601

A heavy uniform cable is used to lift a 300 pound load from ground level to the top of a 100 foot high building. If the cable weighs 20 pounds per linear foot then how much work is done?

Solution

Orient the vertical axis so that the origin is at the top of the building and the positive direction is downward. The work W_L lifting the load is (300)(100) foot-pounds.

```
> W[L] := 300*100*ft*pounds;
```

$$W_L := 30000\ ft\ pounds$$

Since the density of the cable is $\delta = \frac{20\, pounds}{ft}$, the weight of an infinitesimal length of cable of length $dy\, ft$ is $\delta\, dy$. If that length is y feet from the top of the building then the work done on that length is $y\, \delta\, dy$.

The total work W_C lifting the cable is $\int_0^{100\, ft} y\, \delta\, dy$.

```
> delta := 20*pounds/ft;
```

$$\delta := \frac{20\, pounds}{ft}$$

```
> W[C] := Int(y*delta, y = 0 .. 100*ft);
```

$$W_C := \int_0^{100\, ft} \frac{20\, y\, pounds}{ft}\, dy$$

The total work is $W_L + W_C$.

```
> Work = W[L]+value(W[C]);
```

$$Work = 130000\, ft\, pounds$$

Ω

Exercise 7 Page 601

With regard to the preceding exercise, how much work is done in lifting the load from ground level to a height 30 feet above ground level?

Solution

The density δ has the same value assigned in the preceding exercise.

Here there are three components of the total work. There are three components to the total work. The work W_L lifting the load is (300)(30) foot-pounds.

```
> W[L] := 300*pounds*30*ft;
```

$$W_L := 9000\, ft\; pounds$$

The lowest 70 feet of the cable weighs $70\, ft\, \delta$ and is lifted 30 feet. Thus, the work W_{Cable_bottom} lifting the lowest 70 feet of the cable is $70\, ft\, \delta\, 30\, ft$.

```
> W[Cable_bottom] := 70*ft*delta*30*ft;
```

$$W_{Cable\_bottom} := 42000\, ft\; pounds$$

Next consider the top 30 feet of cable. Since the density of the cable is $\delta = \frac{20\; pounds}{ft}$, the weight of an infinitesimal length of cable of length $dy\, ft$ is $\delta\, dy$. If that length is y feet from the top of the building then the work done on that length is $y\, \delta\, dy$. The total work W_{Cable_top} lifting the top 30 feet of cable is

$$\int_0^{30\, ft} y\, \delta\, dy\, .$$

```
> W[Cable_top] := Int(y*delta,y = 0 .. 30*ft);
```

$$W_{Cable\_top} := \int_0^{30\, ft} \frac{20\, y\; pounds}{ft}\, dy$$

The total work is $W_L + W_{Cable\_bottom} + W_{Cable\_top}$.

```
> Work := value(W[L]+W[Cable_bottom]+W[Cable_top]);
```

$$Work := 60000\, ft\; pounds$$

Ω

Exercise 8 Page 601

With regard to the cable and load of Exercise 6, if the load is lifted to the top of the building then how much work is done in lifting it the final (uppermost) 30 feet?

Solution

The work W_L lifting the load is $(300)(30)$ foot-pounds.

```
> W[L] := 300*30*ft*pounds;
```

$$W_L := 9000\, ft\ pounds$$

Since the density of the cable is $\delta = \frac{20\ pounds}{ft}$, the weight of an infinitesimal length of cable of length $dy\, ft$ is $\delta\, dy$. If that length is y feet from the top of the building then the work done on that length is $y\, \delta\, dy$. The total work W_C lifting the cable is $\int_0^{30\, ft} y\, \delta\, dy$. The total work is $W_L + W_C$.

```
> W[C] := Int(y*delta, y = 0 .. 30*ft);
```

$$W_C := \int_0^{30\, ft} \frac{20\, y\ pounds}{ft}\, dy$$

```
> Work = W[L]+value(W[C]);
```

$$60000\, ft\ pounds = 18000\, ft\ pounds$$

Ω

3. Springs: Exercises from the Text

Exercise 9 Page 601

If a spring with spring constant 8 pounds/inch is stretched 7 inches beyond its equilibrium position, then how much work is done?

Solution

In order to use the standard unit foot-pound for work, we convert the measurements in inches to measurements in feet.

```
> k := 8*pound/inch:
  k := subs(inch = ft/12, k);
  F := x -> k*x;
```

$$k := \frac{96\ pound}{ft}$$

$$F := x \rightarrow k\,x$$

```
> Work := Int(F(x), x = 0 .. 7/12*ft);
```

$$Work := \int_0^{\frac{7\ ft}{12}} \frac{96\ pound\ x}{ft}\,dx$$

```
> Work = value(Work);
```

$$\int_0^{\frac{7\ ft}{12}} \frac{96\ pound\ x}{ft}\,dx = \frac{49\ pound\ ft}{3}$$

Ω

Exercise 10, page 601

A spring is stretched 2 inches beyond its equilibrium position. If the force required to maintain it in its stretched position is 60 pounds then how much work has been done?

Solution

In order to use the standard unit foot-pound for work, we convert the measurements in inches to measurements in feet.

```
> restart:
> F := x -> k*x;
```

$$F := x \rightarrow k\,x$$

```
> eqn := 60*pounds = F(2*inch);
```

$$eqn := 60\ pounds = 2\ k\ inch$$

```
> eqn := subs(inch = ft/12, eqn);
```

$$eqn := 60\ pounds = \frac{k\ ft}{6}$$

```
> k := solve(eqn, k);
```

$$k := \frac{360\ pounds}{ft}$$

```
> Work := Int(F(x), x = 0..2/12*ft);
```

$$Work := \int_0^{\frac{ft}{6}} \frac{360\ pounds\ x}{ft}\, dx$$

```
> Work = value(Work);
```

$$\int_0^{\frac{ft}{6}} \frac{360\ pounds\ x}{ft}\, dx = 5\ pounds\ ft$$

Ω

Exercise 11, page 602

A force of 280 pounds compresses a spring 4 inches from its natural length of 16 inches. How much work is done compressing it a further 4 inches?

Solution

In order to use the standard unit foot-pound for work, we convert the measurements in inches to measurements in feet.

```
> restart:
> F := x -> k*x;
```

$$F := x \to k\, x$$

```
> eqn := 280*pounds = F(4*inch);
```

$$eqn := 280\ pounds = 4\ k\ inch$$

```
> eqn := subs(inch = ft/12, eqn);
```

$$eqn := 280\ pounds = \frac{k\ ft}{3}$$

```
> k := solve(eqn, k);
```

$$k := \frac{840\ pounds}{ft}$$

```
> Work := Int(F(x), x = 4/12*ft..8/12*ft);
```

$$Work := \int_{\frac{ft}{3}}^{\frac{2\,ft}{3}} \frac{840\,pounds\,x}{ft}\,dx$$

```
> Work = value(Work);
```

$$\int_{\frac{ft}{3}}^{\frac{2\,ft}{3}} \frac{840\,pounds\,x}{ft}\,dx = 140\,pounds\,ft$$

Ω

Exercise 12, page 602

If 30 foot-pounds of work are done in stretching a spring 3 inches beyond its natural length, then how much work is done stretching it one inch more?

Solution

In order to use the standard unit foot-pound for work, we convert the measurements in inches to measurements in feet.

```
> restart:
> F := x -> k*x;
```

$$F := x \rightarrow k\,x$$

```
> Work := (a,b) -> Int(F(x), x = a.. b);
```

$$Work := (a, b) \rightarrow \int_a^b \mathrm{F}(x)\,dx$$

```
> eqn := value(Work(0*inch, 3*inch)) = 30*ft*pounds;
```

$$eqn := \frac{9\,k\,inch^2}{2} = 30\,ft\,pounds$$

```
> eqn := subs(inch = ft/12, eqn);
```

$$eqn := \frac{k\,ft^2}{32} = 30\,ft\,pounds$$

```
> k := solve(eqn, k);
```

$$k := \frac{960\,pounds}{ft}$$

```
> work := Work(3/12*ft,4/12*ft);
```

$$work := \int_{\frac{ft}{4}}^{\frac{ft}{3}} \frac{960\ pounds\ x}{ft}\, dx$$

```
> work = value(work);
```

$$\int_{\frac{ft}{4}}^{\frac{ft}{3}} \frac{960\ pounds\ x}{ft}\, dx = \frac{70\ ft\ pounds}{3}$$

Ω

Exercise 13, page 602

A spring is stretched a certain length beyond its natural length. What percentage of the total amount of work is expended by the first half of the stretch?

Solution

Let L be the distance the spring is stretched beyond equilibrium. We do not know this value and we do not know the value of the spring constant k. Both parameters will cancel when we form the ratio that gives the required percentage.

```
> restart: F := x -> k*x;
```

$$F := x \rightarrow k\, x$$

```
> percentage := Int(F(x), x = 0..L/2)/Int(F(x), x =
  0..L)*100*percent;
```

$$percentage := \frac{100 \int_{0}^{\frac{L}{2}} k\, x\, dx\ percent}{\int_{0}^{L} k\, x\, dx}$$

```
> value(percentage);
```

$$25\ percent$$

Ω

Exercise 14, page 602

If 2/3 ft-lb of work is done in extending a spring at rest 2 inches beyond its equilibrium position, then how much force is required to maintain the spring in that position?

Solution

In order to use the standard unit foot-pound for work, we convert the measurements in inches to measurements in feet..

```
> restart;
  inch := ft/12;
```

$$inch := \frac{ft}{12}$$

```
> F := x -> k*x;
```

$$F := x \rightarrow k\,x$$

```
> W := (a,b) -> Int(F(x), x=a..b);
```

$$\mathrm{W} := (a, b) \rightarrow \int_a^b \mathrm{F}(x)\,dx$$

```
> eqn := 2/3*ft*pound = W(0*inch,2*inch);
```

$$eqn := \frac{2\,ft\,pound}{3} = \int_0^{\frac{ft}{6}} k\,x\,dx$$

```
> eqn := map(value, eqn);
```

$$eqn := \frac{2\,ft\,pound}{3} = \frac{k\,ft^2}{72}$$

```
> k := solve(eqn, k);
```

$$k := \frac{48\,pound}{ft}$$

```
> F(2*inch);
```

$$8\,pound$$

Ω

Exercise 31, page 603

The magnitude of a force that stretches a spring is given by $\mathrm{F}(x) = k\,x$ when the spring is extended a distance x beyond its equilibrium position. Let $\mathrm{W}(x)$ denote the work done in stretching the spring a distance x from its equilibrium position. Describe the curve that results from plotting the parametric equations $F = \mathrm{F}(x)$, $\mathrm{W} = \mathrm{W}(x)$ in the FW-plane.

Exercise

The spring is *stretched* so x and $F = k\,x$ are positive.

```
> restart; F := x->k*x; eqn1 := F = F(x);
```

$$F := x \to k\,x$$

$$eqn1 := F = k\,x$$

```
> eqn2 := x = solve(eqn1, x);
```

$$eqn2 := x = \frac{F}{k}$$

```
> eqn3 := W = int(F(xi), xi=0..x);
```

$$eqn3 := \mathrm{W} = \frac{k\,x^2}{2}$$

```
> eqn4 := subs(eqn2, eqn3);
```

$$eqn4 := \mathrm{W} = \frac{F^2}{2\,k}$$

The parametric curve is the right half (since $0 \le F$) of a parabola with vertex at the origin of the FW-plane.

Ω

4. Pumping a Fluid from a Reservoir: Exercises from the Text

Exercise 15 page 602

A swimming pool full of water has square base of side 15 feet. It is 10 feet deep and is being pumped dry. The pump floats on the surface of the water and pumps the water to the top of the pool, at which point the water runs off. How much work does the pump perform in emptying the pool?

Solution

A horizontal slice of water that is dy feet thick has weight $62.428\ 15^2\ dy$.
If the slice is y feet from the top, then the work done pumping it to the top of the pool is $y\ 62.428\ 15^2\ dy$.

The total work is $\int_0^{10} y\ 62.428\ 15^2\ dy$.

```
> Work := Int(y*62.428*15^2,y = 0 .. 10);
```

$$Work := \int_0^{10} 14046.300\ y\ dy$$

```
> Work = value(Work);
```

$$\int_0^{10} 14046.300\ y\ dy = 702315.$$

Ω

Exercise 16 page 602

A swimming pool full of water has rectangular base with side lengths of 16 and 30 feet. It is 10 feet deep but not completely filled: the depth of the water it contains is 8 feet. A pump floats on the surface of the water and pumps the water to the top of the pool, at which point the water runs off. How much work does the pump perform in emptying the pool?

Solution

A horizontal slice of water that is dy feet thick has weight $62.428\ 16\ (30)\ dy$.



If the slice is y feet from the top, then the work done pumping it to the top of the pool is $y\ 62.428\ 16\ (30)\ dy$. Because the initial water depth is 8 feet, the slice closest to the top of the pool corresponds to $y = 2$.

The total work is $\int_{2}^{10} y\, 62.428\, 16\, (30)\, dy$.

```
> Work := Int(y*62.428*16*30,y = 2 .. 10);
```

$$Work := \int_{2}^{10} 29965.440\, y\, dy$$

```
> Work = value(Work);
```

$$\int_{2}^{10} 29965.440\, y\, dy = 0.1438341120\ 10^{7}$$

Ω

Exercise 17 page 602

Referring to the swimming pool of the preceding problem, if power is interrupted when half the water has been pumped, then how much work has been done?

Solution

When half the water, initially 8 feet deep, is pumped, 4 feet of water remain. The horizontal slice closest to the top of the pool corresponds to $y = 2$ and the final slice pumped corresponds to $y = 6$ (where y measures the ditance in feet to the top of the tank). The total work is

$\int_{2}^{6} y\, 62.428\, 16\, (30)\, dy$.

```
> Work := Int(y*62.428*16*30,y = 2 .. 6);
```

$$Work := \int_{2}^{6} 29965.440\, y\, dy$$

```
> Work = value(Work);
```

$$\int_{2}^{6} 29965.440\, y\, dy = 479447.0400$$

Ω

Exercise 18 page 602

A semi-cylindrical tank filled with water is 16 feet long. It has rectangular horizontal cross-sections and vertical cross-sections that are semicircular of radius 6 feet. A pump is used which floats on the surface of the tank. It pumps water to the top, at which point the water runs off. How much work is done in pumping the water?

Solution

Orient the positive *y*-axis so that it is directed downward with the origin at the top of the tank.





A "slice" of water that has thickness *dy* and that is y feet from the top of the tank has width $2\sqrt{36-y^2}$, volume $2\,(16)\sqrt{36-y^2}\;dy$, and weight $62.428\;2\,(16)\sqrt{36-y^2}$ **.** The work done in pumping this slice to the top of the tank is $2\,(16)\,62.428\;y\sqrt{36-y^2}\;dy$**.** The total work in pumping the filled tank dry is

$$\int_0^6 2\,(16)\,62.428\;y\sqrt{36-y^2}\;dy.$$

```
> Work := Int(2*16*62.428*y*sqrt(36-y^2),y = 0 .. 6);
```

$$Work := \int_0^6 1997.696\;y\sqrt{36-y^2}\;dy$$

```
> Work = value(Work);
```

$$\int_0^6 1997.696\, y \sqrt{36 - y^2}\, dy = 143834.1120$$

Ω

Exercise 19 page 602

A filled reservoir is in the shape of an inverted cone, with the radius of the base of the cone and the depth at the center both being 100 feet. A pump is used which floats on the surface of the reservoir. It pumps water to the top, at which point the water runs off. How much work is done in pumping the water in the reservoir to a depth of 50 feet?

Solution

Orient the positive y-axis so that it is directed downward with the origin at the top of the tank



Inverted Conical Tank
with height 100 and base radius 100

The diagram below shows a "slice" of water of thickness dy at level y**.** The radius of the slice is $100 - y$**.** The volume of the slice is therefore $\pi\,(100 - y)^2\, dy$ and the weight is $62.428\,\pi\,(100 - y)^2\, dy$**.**



The work done pumping the slice is $y\,62.428\,\pi\,(100-y)^2\,dy$. The slices that are pumped begin at level $y = 0$ and end at level $y = 50$. The total work done is

$$\int_0^{50} y\,62.428\,\pi\,(100-y)^2\,dy.$$

```
> Work := Int(y*62.428*Pi*(100-y)^2, y=0..50);
```

$$Work := \int_0^{50} 62.428\,y\,\pi\,(100-y)^2\,dy$$

```
> Work := value(Work);
```

$$Work := 0.1123623337\;10^{10}$$

Ω

Exercise 20 page 602

A tank filled with water is in the shape of an inverted pyramid with square base. The base of the pyramid measures 25 feet on a side and the height of the pyramid is 40 feet. A pump floats on the surface of the water and pumps water to the upper edge. If the tank is to be emptied so that the remaining water has a depth of 15 feet, then what is the amount of work performed by the pump?

Solution

Orient the positive y-axis so that it is directed downward with the origin at the top of the tank



A horizontal slice of water of thickness *dy* at level y is a square of side length $2x$ where, by similar triangles,

$$\frac{x}{\frac{25}{2}} = \frac{40-y}{40}$$

Thus, $2x = 25 - \frac{5y}{8}$. The volume of the slice is $\left(25 - \frac{5y}{8}\right)^2 dy$ cubic feet. The weight is $62.428\left(25 - \frac{5y}{8}\right)^2 dy$ pounds and the work pumping the slice to the top of the tank is $y\,62.428\left(25 - \frac{5y}{8}\right)^2 dy$ ft-lbs. If the filled tank is emptied so that the remaining water has a depth of 15 feet, then the pumped slice closest to the top is at level $y = 0$ and the pumped slice farthest from the top is at level $y = 40 - 15$, or $y = 25$. The total work done is

$$\int_0^{25} y\,62.428\left(25 - \frac{5y}{8}\right)^2 dy\,.$$

```
> Work := Int(y*62.428*(25-5*y/8)^2,y = 0 .. 25);
```

$$Work := \int_0^{25} 62.428\, y \left(25 - \frac{5\, y}{8}\right)^2 dy$$

```
> Work = value(Work);
```

$$\int_0^{25} 62.428\, y \left(25 - \frac{5\, y}{8}\right)^2 dy = 0.4413600667\ 10^7$$

Ω

Exercise 25 page 602

Suppose that a tank has the shape of a paraboloid of revolution that results from rotating the curve

$$y = 2\, x^2 - 18\ ,\ \text{-3}\ \leq x\ \leq 3$$

about the *y*-axis (*x* and *y* measured in feet). A pump floats on the surface of water and pumps the water to the top of the tank. The tank, initially filled, is pumped until the remaining water is 3 feet deep at the center.
Calculate the work that is done.

Solution

Let $w = -y$**.** The downward-directed axis may be labeled w**.**



Using w instead of y, we may write the rotated curve as $-w = 2\,x^2 - 18$, or $2\,x^2 = 18 - w$, or $x = \sqrt{9 - \frac{w}{2}}$. A horizontal slice w feet from the top of the tank is a disk of radius $x = \sqrt{9 - \frac{w}{2}}$. If the slice has thickness dw, then its volume is $\pi \left(9 - \frac{w}{2}\right) dw$ and its weight is $62.428\,\pi \left(9 - \frac{w}{2}\right) dw$.



The work done in pumping that slice to the top is $62.428\,\pi\, w \left(9 - \frac{w}{2}\right) dw$. The tank is 18 feet deep and initially filled. If it is pumped until 3 feet of water remain, then the levels that have been pumped begin at $w - 0$ and end at $w = 15$. The work done is $\int_0^{15} 62.428\,\pi\,w\left(9 - \frac{w}{2}\right) dw$.

```
> Work := Int(62.428*Pi*w*(9-w/2),w = 0 .. 15);
```

$$Work := \int_0^{15} 62.428\,\pi\,w\left(9 - \frac{w}{2}\right) dw$$

```
> Work = value(Work);
```

$$\int_0^{15} 62.428\,\pi\,w\left(9 - \frac{w}{2}\right) dw = 88255.50578$$

Ω

Exercise 26 page 602

Suppose that a tank has the shape of a paraboloid of revolution that results from rotating the curve

$$y = 2\,x^2 - 18\,,\ \text{-}3\ \le x\ \le 3$$

about the y-axis (x and y measured in feet). A pump floats on the surface of water and pumps the water to the top of the tank. The tank, initially filled, is pumped until the remaining water has half the original volume.
Calculate the work that is done.

Solution

Let $w = -y$**.** The downward-directed axis may be labeled w**.**

In the preceding exercise we calculated x in terms of w**:** $x = \sqrt{9 - \frac{w}{2}}$ **.**

Using the Method of Disks, we find the volume V of the tank to be.

```
> V := Int(Pi*(sqrt(9-w/2))^2, w = 0 .. 18);
```

$$V := \int_0^{18} \frac{\pi(36 - 2w)}{4}\, dw$$

```
> V := value(V);
```

$$V := 81\pi$$

If the tank contains water that has depth d at the center, a level that corresponds to $18 - d$ on the w -axis, then the volume of the water is

```
> Vol := Int(Pi*(sqrt(9-w/2))^2, w = 0 .. 18-d);
```

$$Vol := \int_0^{18-d} \frac{\pi(36 - 2w)}{4}\, dw$$

```
> Vol := value(Vol);
```

$$Vol := -\frac{\pi(18-d)^2}{4} + 9\pi(18-d)$$

If this volume is half the initial volume, then we solve the quadratic equation $Vol = \frac{V}{2}$ for d:

```
> eqn := Vol = V/2;
```

$$eqn := -\frac{\pi(18-d)^2}{4} + 9\pi(18-d) = \frac{81\pi}{2}$$

```
> solve(eqn, d);
```

$$9\sqrt{2}, -9\sqrt{2}$$

We follow the steps of the preceding problem until we come to the upper limit of integration for the work integral. Here, since the depth of the remaining water is $9\sqrt{2}$ instead of 3, we use $18 - 9\sqrt{2}$ for the upper limit instead of $18 - 3$.

```
> Work := Int(62.428*Pi*w*(9-w/2),w = 0 .. 18-9*sqrt(2));
```

$$Work := \int_0^{18-9\sqrt{2}} 62.428\,\pi\, w\left(9 - \frac{w}{2}\right) dw$$

```
> Work = value(Work);
```

$$\int_{0}^{18-9\sqrt{2}} 62.428\,\pi\, w\left(9-\frac{w}{2}\right)dw = 19740.57882$$

Ω

Exercise 27 page 602

Suppose that a tank has the shape of a paraboloid of revolution that results from rotating the curve

$$y = 2x^2 - 18\,,\ -3 \le x \le 3$$

about the y-axis (x and y measured in feet). A pump floats on the surface of water and pumps the water to the top of the tank. The tank, initially filled to a depth of 4 feet at the center, is pumped until it is empty.
Calculate the work that is done.

Solution

Let $w = -y$. The downward-directed axis may be labeled w.
We set up the work integral as in Exercise 25 until we come to the limits of integration. Since the initial depth is 4 feet, the first slice of water we pump is $18-4$ feet from the top of the tank. Thus, our lower limit of integration is 14. We pump all the water, so our upper limit of integration is 18. The work done is $\int_{14}^{18} 62.428\,\pi\, w\left(9-\frac{w}{2}\right)dw$.

```
> Work := Int(62.428*Pi*w*(9-w/2),w = 14 .. 18);
```

$$Work := \int_{14}^{18} 62.428\,\pi\, w\left(9-\frac{w}{2}\right)dw$$

```
> Work = value(Work);
```

$$\int_{14}^{18} 62.428\,\pi\, w\left(9-\frac{w}{2}\right)dw = 12028.89857$$

Ω

Exercise 28 page 602

Suppose that a tank has the shape of a paraboloid of revolution that results from rotating the curve

$$y = 2x^2 - 18 \text{ , } -3 \le x \le 3$$

about the y-axis (x and y measured in feet). A pump floats on the surface of water and pumps the water to the top of the tank. The tank, initially filled to a depth of 4 feet at the center, is pumped until the remaining water is 2 feet deep at the center.
Calculate the work that is done.

Solution

We set up the work integral as in Exerxise 27 until we come to the upper limit of integration. Since we leave 2 feet of water at the center, the last slice of water pumped is $18 - 2$ feet from the top of the tank. Thus, our upper limit of integration is 16. The work done is $\int_{14}^{16} 62.428\,\pi\,w\left(9 - \frac{w}{2}\right) dw$.

```
> Work := Int(62.428*Pi*w*(9-w/2),w = 14 .. 16);
```

$$Work := \int_{14}^{16} 62.428\,\pi\,w\left(9 - \frac{w}{2}\right) dw$$

```
> Work = value(Work);
```

$$\int_{14}^{16} 62.428\,\pi\,w\left(9 - \frac{w}{2}\right) dw = 8760.176129$$

Ω

5. Calculator/Computer Exercises from the Text

Headnote for Exercises 33 - 36

The shape of a reservoir filled to the top with water is obtained by rotating the curve

$$y = 10\left(e^{(-2)} - e^{\left(-\frac{x^2}{2}\right)}\right), \quad 0 \le x \le 2$$

about the *y*-axis. Both x and y are measured in feet. A pump floats on the surface of the water and pumps water to the top, at which point the water runs off.

Exercise 33

How much work is done pumping out a volume of water that leaves the remaining water 4 feet deep at the center?

Solution

The graph of the curve that is rotated (without equal scaling on the axes):



The surface that results from the rotation.



Let $w = -y$. Then the downward-directed vertical axis can be represented by w. The curve of rotation has equation $w = 10\left(\mathbf{e}^{\left(-\frac{x^2}{2}\right)} - \mathbf{e}^{(-2)}\right)$. The positive value of x that corresponds to w is given by $x = \sqrt{-2\ln\left(\mathbf{e}^{(-2)} + \frac{w}{10}\right)}$. Thus, the volume of a horizontal slice that is at level w and that has thickness dw is $\pi\sqrt{-2\ln\left(\mathbf{e}^{(-2)} + \frac{w}{10}\right)}^2\, dw$, or $-2\,\pi\ln\left(\mathbf{e}^{(-2)} + \frac{w}{10}\right) dw$.



The weight of such a slice is $-2\,\pi\,62.428\,\ln\left(\mathbf{e}^{(-2)}+\frac{w}{10}\right)dw$ and the work done pumping the slice to the surface is $-2\,\pi\,62.428\,w\,\ln\left(\mathbf{e}^{(-2)}+\frac{w}{10}\right)dw$. The depth of the tank is $10\,(1-\mathbf{e}^{(-2)})$. If the remaining water is 4 feet deep at the center, then $w=10\,(1-\mathbf{e}^{(-2)})-4$ is the pumped level farthest from the top of the tank. The total work in foot-pounds is

$$-2\,\pi\,62.428\int_0^{10\,(1-\mathbf{e}^{(-2)})-4} w\,\ln\left(\mathbf{e}^{(-2)}+\frac{w}{10}\right)dw$$, or about 3582 ft-lb.

With **Maple**:

```
> f := x -> 10*(exp(-2)-exp(-x^2/2));
```

$$f := x \to 10\,\mathbf{e}^{(-2)}-10\,\mathbf{e}^{(-1/2\,x^2)}$$

```
> eqn1 := y = f(x);
```

$$eqn1 := y = 10\,\mathbf{e}^{(-2)}-10\,\mathbf{e}^{\left(-\frac{x^2}{2}\right)}$$

```
> eqn2 := subs(y=-w, eqn1);
```

$$eqn2 := -w = 10\,\mathbf{e}^{(-2)}-10\,\mathbf{e}^{\left(-\frac{x^2}{2}\right)}$$

```
> eqn3 := x^2 = solve(eqn2, x^2);
```

$$eqn3 := x^2 = -2\,\ln\left(\mathbf{e}^{(-2)}+\frac{w}{10}\right)$$

```
> Work := Int(62.428*Pi*w*x^2, w = 0 .. -f(0)-4);
```

$$Work := \int_0^{6-10\,e^{(-2)}} 62.428\,\pi\,w\,x^2\,dw$$

```
> Work := subs(eqn3, Work);
```

$$Work := \int_0^{6-10\,e^{(-2)}} -124.856\,\pi\,w\,\ln\left(e^{(-2)} + \frac{w}{10}\right) dw$$

```
> Work = evalf( Work );
```

$$\int_0^{6-10\,e^{(-2)}} -124.856\,\pi\,w\,\ln\left(e^{(-2)} + \frac{w}{10}\right) dw = 3582.003545$$

```
> evalf(-2*Pi*62.428*evalf(Int(w*ln(exp(-2)+w/10),w = 0 ..
  10*(1-exp(-2))-4)));
```

$$3582.003546$$

Ω

Exercise 34

How much work is done in pumping out half the water in the reservoir?

Solution

In the preceding exercise we calculated the volume of a horizontal slice at level *w* and thickness *dw* to be $\pi\sqrt{-2\ln\left(e^{(-2)} + \frac{w}{10}\right)}^{\,2}\,dw$, or $-2\,\pi\ln\left(e^{(-2)} + \frac{w}{10}\right) dw$. The volume of the tank is therefore $-2\,\pi\int_0^{10\,(1-e^{(-2)})} \ln\left(e^{(-2)} + \frac{w}{10}\right) dw$, or about 37.322 cubic feet. Half the water is 18.661 cubic feet.

Next we find the level that corresponds to this quantity of water. Doing so requires us to solve the equation $-2\,\pi\int_0^{h} \ln\left(e^{(-2)} + \frac{w}{10}\right) dw = 18.661$ for h. We solve this equation numerically (using Maple's fsolve) and find $h = 2.0117$. The work done is therefore

$$\int_0^{2.0117} -124.856\,\pi\,w\,\ln\left(\mathbf{e}^{(-2)} + \frac{1\,w}{10}\right) dw$$, or 1054.5 ft-lb.

With **Maple**:

```
> evalf(-2*Pi*Int(ln(exp(-2)+w/10),w = 0 .. 10*(1-exp(-2))));
```

$$37.32175318$$

```
> eqn := -2*Pi*int(ln(exp(-2)+w/10),w = 0 .. h) = 18.661;
```

$$eqn := -2\,\pi\left(10\ln\left(\mathbf{e}^{(-2)} + \frac{h}{10}\right)\mathbf{e}^{(-2)} + \ln\left(\mathbf{e}^{(-2)} + \frac{h}{10}\right)h + 20\,\mathbf{e}^{(-2)} - h\right) = 18.661$$

```
> eqn_for_h := h = fsolve(eqn, h, 0 .. evalf(10*(1-exp(-2))));
```

$$eqn\_for\_h := h = 2.011679439$$

```
> Work := subs(eqn_for_h, Int(-124.856*Pi*w*ln(exp(-2)+1/10*w),w
= 0 .. h));
```

$$Work := \int_0^{2.011679439} -124.856\,\pi\,w\ln\left(\mathbf{e}^{(-2)} + \frac{w}{10}\right) dw$$

```
> Work = value(Work);
```

$$\int_0^{2.011679439} -124.856\,\pi\,w\ln\left(\mathbf{e}^{(-2)} + \frac{w}{10}\right) dw = 1054.520077$$

Ω

Exercise 35

How much work is done in pumping out all the water in the reservoir?

Solution

The solution is the same as for Exerxise 33, except that the upper limit of integration is $10\,(1 - \mathbf{e}^{(-2)})$ i. The total work in foot-pounds is $-2\,\pi\,62.428\int_0^{10\,(1-\mathbf{e}^{(-2)})} w\ln\left(\mathbf{e}^{(-2)} + \frac{w}{10}\right) dw$, or about 5754.9 ft-lb.

```
> evalf(-2*Pi*62.428*int(w*ln(exp(-2)+w/10),w = 0 ..
  10*(1-exp(-2))));
```

$$5754.929127$$

Ω

Exercise 36

In pumping out all the water in the reservoir, how deep is the remaining water at the center when half the work is done?

Solution

From the preceding exercise, half the work is 2877.464564 ft-lb. We must solve the equation

$$-2\,\pi\,62.428\int_0^h w\ln\left(\mathbf{e}^{(-2)}+\frac{w}{10}\right)dw = 2877.464564 \quad \text{for } h.$$

```
> eqn := -2*Pi*62.428*Int(w*ln(exp(-2)+w/10),w = 0 .. h) =
  2877.464564;
```

$$eqn := -124.856\,\pi\int_0^h w\ln\left(\mathbf{e}^{(-2)}+\frac{w}{10}\right)dw = 2877.464564$$

```
> eqn := map(value, eqn);
```

$$eqn := -124.856\,\pi\left(-50\ln\left(\mathbf{e}^{(-2)}+\frac{h}{10}\right)\left(\mathbf{e}^{(-2)}\right)^2+\frac{1}{2}\ln\left(\mathbf{e}^{(-2)}+\frac{h}{10}\right)h^2+75\left(\mathbf{e}^{(-2)}\right)^2+5\,\mathbf{e}^{(-2)}\,h-\frac{h^2}{4}-175\,\mathbf{e}^{(-4)}\right) = 2877.464564$$

```
> eqn_for_h := h = fsolve(eqn, h, 0..evalf(10*(1-exp(-2))));
```

$$eqn\_for\_h := h = 3.913887267$$

The depth is

```
> evalf( subs(eqn_for_h, 10*(1-exp(-2)) - h ) );
```

$$4.732759901$$

Ω

Code for Figures

Code

Copyright and Author Information

Copyright and Author Information
BlankKrantz-08_5R8.mws   A Maple Release 8 worksheet.

Author: Brian E. Blank (12 November 2007)

For more information, please contact the author:

Department of Mathematics,
Washington University in St. Louis
St. Louis, MO   63130

Telephone: (314) 935-6763
e-mail: brian@math.wustl.edu